
COMP1130: Lambda Calculus - Worksheet 3

� Harder Beta Reduction

Reduce the following terms to normal form (apply as many beta reductions as possible). If you
are comfortable with alpha conversions you do not need to write them out now.

1. (λx.xx)(λy.y)

Solution.
(λx.xx)(λy.y) →

β
(λy.y)(λy.y) →

β
λy.y

2. (λx.λy.yxy)(xx)z

Solution.
(λx.λy.yxy)(xx)z →

β
(λy.y(xx)y)z →

β
z(xx)z

3. (λx.λy.x)(λx.λy.y)w

Solution.
(λx.λy.x)(λx.λy.y)w →

β
(λy.(λx.λz.z))w →

β
λx.λz.z

4. (λx.(λx.x)(λz.zy)x)(λz.zw)

Solution.

(λx.(λx.x)(λz.zy)x)(λz.zw)

→
β
(λx.x)(λz.zy)(λz.zw)

→
β
(λz.zy)(λz.zw)

→
β
(λz.zw)y

→
β
yw

5. (λx.xx)(λy.λz.y)

Solution.
(λx.xx)(λy.λz.y) →

β
(λy.λz.y)(λy.λz.y) →

β
λx.λy.λz.y

6. (λx.xxx)(λy.yw)

Solution.

(λx.xxx)(λy.yw)

→
β
(λy.yw)(λy.yw)(λy.yw)

→
β
(λy.yw)w(λy.yw)

→
β
ww(λy.yw)

� Encoding Pairs

Recall that

(a, b) := λx.if x then a else b

fst = λp.p true

snd = λp.p false

1. Find a lambda expression for the function pairmap, that applies a function to each element
of a pair, that is

pairmap f (a, b) →
β

(fa, fb)

Solution. The solution to this is to write a function that extracts out each part of the
pair, applies f to it, and then packs the two parts back together in a pair

pairmap (a, b) = (fa, fb)

Replacing (a, b) with a generic variable p, we would then need to use fst and snd to extract
each part out.

pairmap p = (f(fst p), f(snd p))

We can now use abstraction to move the p over,

pairmap = λp.(f(fst p), f(snd p))

and then replace each function we used to build pairmap with it’s definition

pairmap = λp.(f((λq.q true)p), f((λq.q false)p))

→
β

λp.(f(p true), f(p false))

= λp.(f(p(λx.λy.x)), f(p(λx.λy.y)))

= λp.λx.if x then f(p(λx.λy.x)) else f(p(λx.λy.y))

= λp.λx.x(f(p(λx.λy.x)))(f(p(λx.λy.y)))

2. Find a lambda expression for the function swap, that swaps the elements of a pair, that is,

swap (a, b) →
β

(b, a)

Solution. We can use the same construction as before, but swap the position of fst and
snd, and remove the application of the function f .

After doing so, we obtain as the final result

swap = λp.λx.x(pλx.λy.y)(pλx.λy.x)

3. What could be a sensible definition of a 3-tuple (a, b, c)?

Solution. One option is to nest the pairs that already exist.

(a, b, c) := ((a, b), c)

4. Based on your definition above, try to define the functions fst3, snd3, trd3 such that

fst3(a, b, c) →
β

a

snd3(a, b, c) →
β

b

trd3(a, b, c) →
β

c

Solution. Since my 3-tuples are just made out of pairs underneath, it’s clear how to define
these using fst and snd.

fst3 = λp.(fst(fst p))

snd3 = λp.(snd(fst p))

trd3 = snd

5. Define a function rot3 such that rot3(a, b, c) = (b, c, a).

Solution. We can use the above definitions to do this.

rot3 = λp.(snd3 p, trd3 p, fst3 p)

� Encoding Numerals (Church)

Recall from lectures that the Church encoding of natural numbers is

0 :=λf.λx.x

1 :=λf.λx.fx

2 :=λf.λx.f(fx)

. . .

n :=λf.λx.fn(x)

succ :=λn.λf.λx.f(nfx)

iszero :=λz.z(λy.false)true

where fn(x) is shorthand for the function f applied to x, n times over.

1. Verify that iszero 0 →
β
true and iszero 1 →

β
false

Solution.

iszero 0 = (λz.z(λy.false)true)(λf.λx.x)

→
β

(λf.λx.x)(λy.false)true

→
β

(λx.x)true

→
β
true

iszero 1 = (λz.z(λy.false)true)(λf.λx.fx)

→
β

(λf.λx.fx)(λy.false)true

→
β

(λx.(λy.false)x)true

→
β

(λy.false)true

→
β
false

2. Verify that succ 0 →
β
1 and succ 1 →

β
2.

Solution.

succ 0 = (λn.λf.λx.f(nfx))(λf.λx.x)

→
β

λf.λx.f((λf.λx.x)fx)

→
β

λf.λx.f((λx.x)x)

→
β

λf.λx.fx = 1

succ 1 = (λn.λf.λx.f(nfx))(λf.λx.fx)

→
β

λf.λx.f((λf.λx.fx)fx)

→
β

λf.λx.f((λx.fx)x)

→
β

λf.λx.f(fx) = 2

3. Suppose you were given a function1 pred that satisfied the property that

pred n =

n-1 n ̸=
α
0

0 n =
α
0

for any natural number n.

Using pred and previously defined functions, try to define the isOne function satisfying

isOne n =

true n =
α
1

false n ̸=
α
1

Solution. The easiest way to do this is to first check if the number is zero, and return
false if so. If the number is not so, decrement, and then return the result obtained when
checking if the new number is zero. Formally,

isOne = λn.if (iszero n) then false else iszero(pred n)

4. Check that succ actually does what we claim, that succ n = n+1. (Hint: Try to evaluate
succ given an arbitrary Church numeral n as input, and use the fact that f(fn(x)) =
fn+1(x).)

Solution.

succ n = (λn.λf.λx.f(nfx))

→
β

(λn.λf.λx.f(nfx))(λf.λx.fn(x))

→
β

λf.λx.f((λf.λx.fn(x))fx)

→
β

λf.λx.f((λx.fn(x))x)

→
β

λf.λx.f(fn(x))

→
β

λf.λx.(fn+1(x)) = n+1

5. Note that our definition of numbers is a function that takes two arguments, and applies the
first to the second, many times over. With that in mind, let foo := λn.2 succ n. What do
you think the foo function does when given Church numerals as input?

Solution. We can expand out the definition of foo,

foo := λn.2 succ n

= λn.(λf.λx.f(fx))succ n

→
β

λn.(λx.succ (succ x))n

→
β

λn.succ (succ n) = n+2

It looks like foo takes a natural number and adds 2 to it.
1You can try to derive the definition of pred yourself, but it’s very difficult!

6. Using the previous exercise as a starting point, try to define the function plus that satisfies
the property plus n m = n+m for any Church numerals n,m.

Solution. We can generalise the previous example by replacing 2 with one of the inputs to
the function.

plus = λn.λm.n succ m

7. Define bar := λn.n (plus 2) 0. What do you think that bar does when given Church
numerals as input?

Solution. We already know that natural numbers act to apply a function many times to
an argument. Consider the case where bar is given an input of 3. (We can also use the
known properties of plus to ease evaluation.) Then,

(λn.n (plus 2) 0) 3

→
β

(λf.λx.f(f(f(x)))) (plus 2) 0

→
β

(λx.(plus 2)((plus 2)((plus 2)x))) 0

→
β

(plus 2)((plus 2)((plus 2)0))

→
β

(plus 2)((plus 2)(0+2))

→
β

(plus 2)((plus 2)(2))

→
β

(plus 2)(2+2)

→
β

(plus 2)4

→
β
4+2 = 6

It looks like bar takes a natural number and doubles it.

8. Using the previous exercise as a starting point, try to define the function mul that satisfies
the property mul n m = n*m for any Church numerals n,m.

Solution. By modifying the above example function bar, we can replace the 2 with another
input variable, and obtain

mul = λn.λm.n (plus m)0

9. Try and define a function pow with the property that

pow n m = nm

for any natural numbers n,m.

Solution. The construction is similar to building mul from plus, by noting that nm is the
same as n multiplied by itself, m times over. Also note that 1 is the multiplicative identity,
so that will be our base case to start from.

pow = λn.λm.m (mul n) 1

� Writing functions with a Fixed Point Combinator

To write functions like the factorial function, we write an intermediate function that has the
body of the factorial function, but it takes the function to call again as an argument

F := λf.λn. if (isZero n) then 1 else (mul n (f (pred n)))

We then use any fixed point combinator (for example, Turing’s fixed point combinator Θ) to
give us the recursion, as it will call F over and over again.

fac = ΘF

1. Verify that fac 3 = 6. (You may assume that Θ, isZero, mul, pred ect. do what they
should do, rather than evaluating the whole thing tediously by hand.)

Solution.

fac 3

=ΘF3

→
β
F (ΘF )3

=(λf.λn. if (iszero n) then 1 else (mul n (f (pred n)))(ΘF )3

→
β

if (iszero 3) then 1 else (mul 3 (ΘF (pred 3)))

→
β

if (false) then 1 else (mul 3 (ΘF (pred 3)))

→
β
mul 3 (ΘF 2)

→
β
mul 3 (F (ΘF ) 2)

→
β
mul 3 (if (iszero 2) then 1 else (mul 2 (ΘF (pred 2)))

→
β
mul 3 (mul 2 (ΘF (pred 2)))

→
β
mul 3 (mul 2 (mul 1 (ΘF 0)))

→
β
mul 3 (mul 2 (mul 1 (F (ΘF ) 0)))

→
β
mul 3 (mul 2 (mul 1 (if (iszero 0) then 1 else (mul 3 (ΘF (pred 0))))))

→
β
mul 3 (mul 2 (mul 1 (if (true) then 1 else (mul 3 (ΘF (pred 0))))))

→
β
mul 3 (mul 2 (mul 1 1)) →

β
6

With the help of previously defined functions (like succ, pred, isZero), try to write the fol-
lowing functions with help of Θ.
(Hint: It might help to write the Haskell definition first!)

1. Define sum with the property that for all natural numbers n representing values of n, that

sum n = 1 + 2 + . . .+ n

Solution.

We first write it in Haskell, to identify the body of the code.

sum :: Int -> Int

sum n

|n == 0 = 0

|otherwise = n + sum (n-1)

This indicates we should use

G = λf.λn.if (iszero n) then 0 else (plus n (f (pred n)))

and then define sum using the Θ combinator, as before

sum = ΘG

2. Define geq with the property that

geq n m =

{
true n ≥ m

false otherwise

for all natural numbers n,m representing values n,m respectively. (Note that this is given
in lecture slides; you could try to do it without consulting the slides. Can you give a more
efficient solution?)

Solution.

geq :: Int -> Int -> Int

geq n m

|n == 0 && m == 0 = True

|n == 0 = False

|m == 0 = True

|otherwise = geq (n-1) (m-1)

G = λf.λn.λm.if (and (iszero n) (iszero m)) then true else(

if (iszero n) then false else(

if (iszero m) then true else(

f (pred n) (pred m))))

geq = ΘG

3. Define strgeq with the property that

strgeq n m =

{
true n > m

false otherwise

for all natural numbers n,m representing values n,m respectively.
(Hint: you may be able to use functions you have written previously here, along with the
equals function from the lecture.)

Solution. Note that
(n > m) ≡ (n ≥ m) ∧ (n ̸= m)

Hence,
strgeq = λn.λm.and (geq n m) (not (eq n m))

4. Define sub to compute saturated subtraction, that is, for any natural numbers n,m repre-
senting values n,m we have that

sub n m =

{
0 n < m

n−m n ≥ m

Solution.

Notice that n−m = (pred n)− (pred m), where so we can use this to define sub in terms
of pred.

sub :: Int -> Int -> Int

sub n m

|m == 0 = n

|otherwise = sub (pred n) (pred n)

and convert, as before,

G = λf.λn.λm.if (iszero m) then n else( f (pred n) (pred m))

5. (Tricky) Define quotRem to compute integer division with remainder 2. Formally, it should
have the property that quotRem returns a pair of natural numbers

quotRem n d = (q, r)

satisfying the property that qd+ r = n.
(Hint: You might need to write a “helper” function with an extra parameter to keep track
of the state of q during computation, as you would in Haskell.)

Solution.

As the hint suggests, we need a helper function to keep track of the current quotient q.
We implement division via repeated subtraction, and stop when the dividend n is strictly
smaller than the divisor d. We return a tuple of both the quotient q and the remainder r.

quotRem :: Int -> Int -> (Int,Int)

quotRem n d = helper 0 n d

where

helper :: Int -> Int -> Int -> (Int,Int)

helper q n d

|d > n = (q,n)

|otherwise = helper (q+1) (n-d) d

We define

helper := λf.λq.λn.λd if (strgeq d n) then (q, n) else (f (succ q) (sub n d) d)

and then use the Θ combinator to give us the recursion, initializing the value of q to zero.

quotRem := λn.λd(Θ helper 0 n d)

2The name quotRem comes from “quotient” and “remainder”. The function is also defined already in the Haskell
prelude.

